
1 Objective

The objective of this homework is to give initial exposure to the circuit simulation software environment that will be used in this course throughout the semester.  The Simulation Program with Integrated Circuit Emphasis (SPICE) was developed at UC Berkeley and is one of the most recognized circuit simulation programs.  There are numerous books and web resources available for help using this program 

The supported simulation software for this class is HSPICE (netlist simulator) and its corresponding waveform analyzer, Awaves, both commercial implementations from Synopsys (formerly Avant!).  Both of these programs are available on the Berkeley servers, to which all students in the course will receive accounts.  If you cannot or do not want to use the Berkeley versions, you are free to use any SPICE software you have available to you but we may not be able to help you if you encounter problems.

2 Getting started

2.1 Verify your account is setup correctly

The default account setup for the class should allow you to run HSPICE without any further configuration.  To verify this, type:


which hspice

If it doesn’t give an error like ‘Command not found’, then you should be ready to go.  Additionally, SPICE creates a number of data files, so you should make a separate directory for it.  You may want something like


mkdir hw1


cd hw1

2.2 Describing a circuit in SPICE

In this homework, you will simulate the circuit in Figure 1.1.  This circuit is a simple Resistor-Transistor Logic (RTL) inverter using a Bipolar Junction Transistor (BJT).  As an aside, this is probably the last time you will see a BJT in this class, since we will focus on MOSFET-based logic.  In addition to the BJT labeled Q1, there are two resistors, RB and RC, and two independent voltage sources, the circled VCC and VIN.

The first step toward describing a circuit in SPICE is to label all the nodes.  A node is any point in the circuit where two elements are connected.  Nodes can be numbers or names, with the only restriction being that the circuit ground should be labeled 0 (ie. the number zero).  As shown in the figure, the nodes are labeled: 0, in, base, vcc, and out.

A SPICE file has four main components: a title, a set of models, a netlist, and a list of analysis commands.  The title is always the first line of the file.  Comments can be included elsewhere in the file by prefacing the line with an asterisk (*).

The syntax of a SPICE netlist is simple.  Just include a line for each element of the circuit.  The format of this line depends on the type of element it is.  Here’s a quick list of the most common circuit elements and their simplified syntax (note that the plus sign is used to indicate the line is continued from the previous one):

Resistor
Rname +node –node value
Capacitor
Cname +node –node value [IC=initial_voltage]

Inductor
Lname +node –node value [IC=initial_current]

Bipolar transistor
Qname collector_node base_node emitter_node model_name
MOSFET
Mname drain_node gate_node source_node bulk_node model_name 
+ [L=value] [W=value]

Independent voltage source
Vname +node –node [[DC value] [AC magnitude [phase]] [transient_value]

The nodes are just the arbitrary names we picked earlier, but node connections are not enough to uniquely define some circuit elements.  For example a BJT also has characteristics like NPN or PNP type, beta, Early voltages, saturation currents, etc.  Model cards are separate descriptions of a particular kind of element that hold these characteristics.  A model for the BJT is available in ~msheets/pub/ee141/npn.mod and also on the web site.  In order to use this model, it has to be included in our SPICE deck (the terms card and deck are throwbacks to the days of punch cards).

The transient_value input to a point source can take many forms, the most common of which is a signal pulse.  The syntax for this is


PULSE( V1 V2 [td [tr [tf [pw [period]]]]] )

where V1 is the initial voltage, V2 is the peak voltage, td is the initial delay time, tr is the rise time, tf is the fall time, pw is the pulse width, and period is the pulse period.  We will use the PULSE command in the VIN voltage source to simulate a near-square wave input.

The analysis section is used to tell SPICE what you want to simulate.  The three main analysis types that are used most often are AC, DC, and transient.  An AC analysis is used to examine the behavior of a circuit with frequency as the x-axis.  For example, a filter transfer function could be simulated by sweeping from 1 Hz to 1 MHz and observing the Vout/Vin relationship.  A DC analysis is used to examine the bias point of a circuit when swept over a variable.  For example, the output of a circuit could be simulated as the supply voltage is changed from 2.5 volts to 5 volts or as the temperature varies from 70 degrees C to 100 degrees C.  Lastly, the transient analysis is used to examine the behavior of a circuit over a period of time.  This will be the most common case in this course.

For this circuit, we will examine what happens in the first 30 ns of operation (saving data every 1 ns) when the input pulse on VIN is applied (transient analysis).  We will also see what happens when the input is a voltage ramp from 0 to 5 volts with a step increment of 0.1 volts (DC analysis).  Note that in a DC analysis the PULSE waveform is ignored, since it is a transient value.

The extra control information specifies how to write the output files (.options post) and to write the operating point of the circuit (.op) into the log file.

Below is the entire SPICE file for this homework.  Use any text editor (vi, pico, emacs, etc.) to enter it into a file called RTLinv.sp.  Alternatively, you may copy the file 

cp ~msheets/pub/ee141/RTLinv.sp .

already typed in, but please make sure you understand all the sections.  If you are not using the Berkeley server, you also need to download the npn.mod card from the web site.

RTLinv.sp:

Simple RTL Inverter

.include ‘~msheets/pub/ee141/npn.mod’

*netlist---------------------------------------

VCC vcc 0 5

VIN in 0 PULSE 0 5 2NS 2NS 30NS 60NS

RB in base 10k

Q1 out base 0 NPN

RC vcc out 1k

*extra control information---------------------

.options post=2 nomod

.op

*analysis--------------------------------------

.TRAN 1ns 30ns

.DC VIN 0 5 0.1

.END

2.3 Simulating the circuit
HSPICE is run from the command line as follows:

hspice RTLinv.sp >! RTLinv.out

Upon proper completion of the simulation, you should see the following:

info:
***** hspice job concluded

If it instead says the job was aborted, you should use a text editor (emacs, vi, more, etc) to examine the log file RTLinv.out for an error message.  Correct the error, and then rerun the HSPICE command.

HSPICE creates a number of output files in the same directory as the SPICE deck:

RTLinv.sp is the input netlist (your file)

RTLinv.sw0 is the DC sweep data output. (used by awaves)

RTLinv.tr0 is the transient data output. (used by awaves)

RTLinv.out is the output listing from HSPICE. (look here for errors or text about the circuit)

RTLinv.st0 is the simulation run information. (not useful)

RTLinv.ic is the information about input to HSPICE (not useful)

2.4 Viewing the results of the transient analysis

A separate program is used to graphically observe the output of SPICE.  This program uses the Xwindows interface, so you must have an Xserver running on your local machine and have properly configured your user account.  Start awaves by entering the following:

awaves &
Once awaves loads, click on Design/Open.  This will open a menu to select which netlist file to display.  Your netlist RTLinv.sp should be listed, otherwise, switch to the correct directory using the tab or the arrows.  Once you have found your netlist, double-click on RTLinv.sp, which should open the Results Browser.

To view the transient waveforms, click on Transient: Simple RTL Inverter and either double click on the waveform you wish to see or right click on the waveform and drag the wave with the center mouse button to the panel you want to display the waveform.

Print the waveforms for in and out.

2.5 Viewing the results of the DC analysis

Open a new panel for the DC waveform by clicking on Panels/Add.  Return to the Results Browser or reopen by clicking Tools/Results Browser.  Click on DC: Simple RTL Inverter and display the waveform for out.

2.6 Transforming the RTL gate into a pseudo-NMOS inverter

In this section, the bipolar transistor will be replaced with a MOS transistor with L=0.25u and W= 100u (refer to Figure 1.2)

First, include a model for a MOSFET transistor by replacing the “.include” line with the g25.mod library:

.lib ‘g25.mod’ TT

A library is a collection of models.  The TT on the end of the line specifies to use the Typical NMOS and Typical PMOS model for the transistors.  Also available in this same library are Slow versions (SS), Fast versions (FF), and mixtures (SF) and (FS).  For the purposes of this class, we will use only the TT version.  In this library, NMOS transistors are named “N” and PMOS transistors are named “P”.  Further edit MOSinv.sp to replace the BJT with a MOSFET:

Delete:

Q1 out base 0 npn
Replace with:
M1 out base 0 0 N L=0.25u W=100u
Be sure to copy the g25.mod file to your working directory.  You can download it from the web page or copy it on the Berkeley servers:


cp ~msheets/pub/ee141/g25.mod .

2.7 Simulation and Analysis of the MOS inverter

Repeat the analysis steps 2.3-2.5, this time using MOSinv.sp as the input file.

3 Report

For your report, all that is required is the following: 

· A printout of the SPICE input files

· Plot of the transient response of the pseudo-NMOS inverter

· Plot of the DC transfer characteristic of the pseudo-NMOS inverter

Be sure to label your plots (either edit the title before printing in Awaves or just write it by hand).

